
Equivalencia lógica

x F y G son equivalentes si |= F ↔ G.

x Decisión de la equivalencia:

u es equivalente(+F,+G) se verifica si las fórmulas F y G son equivalentes.

u Ejemplos:

?- es_equivalente(-(p & q), -p v -q).
Yes
?- es_equivalente(-(p & q), -p & -q).
No

u Def. de es equivalente:

es_equivalente(F,G) :-
es_tautologia(F <=> G).

Forma normal negativa

x Forma normal negativa:

u Una fórmula está en forma normal negativa (FNN) si no contiene las conectivas

→, ↔ y la negación no se aplica a subfórmulas compuestas.

u Ejemplos:

• (¬p ∨ q) ∧ (¬q ∨ p) está en FNN

• (p → q) ∧ (q → p) no está en FNN

• ¬(p ∧ q) no está en FNN

u Una fórmula G es una forma normal negativa (FNN) de la fórmula F si G está

en forma normal negativa y es equivalente a F .

u Ejemplo: una FNN de p ↔ q es (¬p ∨ q) ∧ (¬q ∨ p).

Cálculo de forma normal negativa

u Aplicando a una fórmula F los siguientes pasos se obtiene una forma normal nega-

tiva de F :

1. Eliminar las equivalencias usando la relación

A ↔ B ≡ (A → B) ∧ (B → A) (1)

2. Eliminar las implicaciones usando la equivalencia

A → B ≡ ¬A ∨ B (2)

3. Interiorizar las negaciones usando las equivalencias

¬(A ∧ B) ≡ ¬A ∨ ¬B (3)

¬(A ∨ B) ≡ ¬A ∧ ¬B (4)

¬¬A ≡ A (5)

Cálculo de forma normal negativa

u fnn(+F,?G) se verifica si G es una forma normal negativa de la fórmula F.

u Ejemplos:

?- fnn(p <=> q,F).
F = (-p v q) & (-q v p)
?- fnn(p v -q => r, F).
F = (-p & q) v r
?- fnn(p & (q => r) => s, F).
F = (-p v (q & -r)) v s

u Def. de fnn:

fnn(F, G) :-
elimina_equivalencias(F,F1),
elimina_implicaciones(F1,F2),
interioriza_negaciones(F2,G).

Cálculo de forma normal negativa

u elimina equivalencias(+F,?G) se verifica si G es la fórmula obtenida eliminando las

equivalencias de la fórmula F, usando la reducción (1).

u Def. de elimina equivalencias:

elimina_equivalencias(A <=> B,
(A1 => B1) & (B1 => A1)) :- !,

elimina_equivalencias(A, A1),
elimina_equivalencias(B, B1).

elimina_equivalencias(A, B) :-
A =.. [Op|L1], !,
maplist(elimina_equivalencias,L1,L2),
B =.. [Op|L2].

elimina_equivalencias(A, A).

Cálculo de forma normal negativa

u elimina implicaciones(+F,?G) se verifica si G es la fórmula obtenida eliminando las

implicaciones de la fórmula F, usando la reducción (2).

u Def. de elimina implicaciones:

elimina_implicaciones(A => B, -A1 v B1) :- !,
elimina_implicaciones(A, A1),
elimina_implicaciones(B, B1).

elimina_implicaciones(A, B) :-
A =.. [Op|L1], !,
maplist(elimina_implicaciones,L1,L2),
B =.. [Op|L2].

elimina_implicaciones(A, A).

Cálculo de forma normal negativa

u interioriza negaciones(+F,?G) se verifica si G es la fórmula obtenida interiorizando

las negaciones de la fórmula F (que no tiene => ni <=>) de forma que las negaciones

se apliquen sólo sobre śımbolos proposicionales. Las reglas de interiorización son

las reduciones (3), (4) y (5).

u Def. de interioriza negaciones:

interioriza_negaciones(-(A & B), A1 v B1) :- !,
interioriza_negaciones(-A, A1),
interioriza_negaciones(-B, B1).

interioriza_negaciones(-(A v B), A1 & B1) :- !,
interioriza_negaciones(-A, A1),
interioriza_negaciones(-B, B1).

interioriza_negaciones((-(-A)), A1) :- !,
interioriza_negaciones(A, A1).

interioriza_negaciones(A, B) :-
A =.. [Op|L1], !,
maplist(interioriza_negaciones,L1,L2),
B =.. [Op|L2].

interioriza_negaciones(A, A).

Forma normal conjuntiva

x Forma normal conjuntiva:

u Una fórmula está en forma normal conjuntiva (FNC) si es una conjunción de

disyunciones de literales; es decir, es de la forma

(L1,1 ∨ . . . ∨ L1,n1) ∧ . . . ∧ (Lm,1 ∨ . . . ∨ Lm,nm).

u Ejemplos:

• (¬p ∨ q) ∧ (¬q ∨ p) está en FNC

• (¬p ∨ q) ∧ (q → p) no está en FNC

u Una fórmula G es una forma normal conjuntiva (FNC) de la fórmula F si G está

en forma normal conjuntiva y es equivalente a F .

u Ejemplo: Una FNC de ¬(p ∧ (q → r)) es

(¬p ∨ q) ∧ (¬p ∨ ¬r).

Cálculo de forma normal conjuntiva

u Aplicando a una fórmula F los siguientes pasos se obtiene una forma normal con-

juntiva de F :

1. Calcular una forma normal negativa de F .

2. Interiorizar las disyunciones usando la propiedad distributiva de la disyunción

sobre la conjunción

A ∨ (B ∧ C) ≡ (A ∨ B) ∧ (A ∨ C) (6)

(A ∧ B) ∨ C ≡ (A ∨ C) ∧ (B ∨ C) (7)

u fnc(+F,?G) se verifica si G es una forma normal conjuntiva de la fórmula F.

u Ejemplos:

?- fnc(p & (q => r), F).
F = p& (-q v r)
?- fnc(-(p & (q => r)), F).
F = (-p v q)& (-p v-r)

u Def. de fnc:
fnc(F,G) :-

fnn(F,F1),
interioriza_disyunciones(F1,G).

Cálculo de forma normal conjuntiva

interioriza_disyunciones(A v (B & C), ABC) :- !,
interioriza_disyunciones(A v B, AB),
interioriza_disyunciones(A v C, AC),
interioriza_disyunciones(AB & AC, ABC).

interioriza_disyunciones((A & B) v C, ABC) :- !,
interioriza_disyunciones(A v C, AC),
interioriza_disyunciones(B v C, BC),
interioriza_disyunciones(AC & BC, ABC).

interioriza_disyunciones(A, B) :-
A =.. [Op|L],
maplist(interioriza_disyunciones,L,L1),
( L1 = L ->

B =.. [Op|L1]
; % not(L1 = L) ->

B1 =.. [Op|L1],
interioriza_disyunciones(B1,B)).

interioriza_disyunciones(A, A).

Transformación a cláusulas

x Cláusulas
u Una cláusula es un conjunto de literales.

u Variables sobre cláusulas: C, C1, C2, . . .

u El valor de una cláusula C en una interpretación I es

I(C) =

{
1, si existe un L ∈ C tal que I(L) = 1

0, en caso contrario.

u Una fórmula F y una cláusula C son equivalentes si I(F ) = I(C) para cualquier

interpretación I.

u Si C = {L1, L2, . . . , Ln}, entonces C es equivalente a L1 ∨ L2 ∨ . . . ∨ Ln.

u La cláusula vaćıa es el conjunto vaćıo de cláusulas y se representa por ¤.

u En cualquier interpretación I, I(¤) = 0.

Transformación a cláusulas

x Fórmulas clausales
u Las fórmulas clausales son las fórmulas obtenidas mediante las siguientes reglas

• Si F es un literal, entonces F es una fórmula clausal

• Si F y G son fórmulas clausales, entonces F ∨ G es una fórmula clausal.

u Ejemplos:

• p, ¬p y ¬p ∨ (q ∨ ¬r) son fórmulas clausales

• ¬p ∨ (q ∧ ¬r) no es una fórmula clausal

Transformación a cláusulas

x Transformación de fórmulas clausales en cláusulas
u El siguiente procedimiento transforma fórmulas clausales en cláusulas equivalentes:

Cláusula(F ) =

=

{ {F}, si F es un literal;

Cláusula(F1) ∪ Cláusula(F2), si F = (F1 ∨ F2)

u clausula(+F,-C) se verifica si C es una cláusula equivalente a la fórmula clausal F

u Ejemplos:

?- clausula(p,C). => C = [p]
?- clausula(-p,C). => C = [-p]
?- clausula((-p v r) v (-p v q),C). => C = [q, r, -p]

u Def. de clausula:

clausula(F1 v F2, S) :- !,
clausula(F1, S1),
clausula(F2, S2),
append(S1, S2, S3),
sort(S3,S).

clausula(L, [L]).

PL 2002–03 CcIa Implementación en Prolog de la transformación a cláusulas 6.14



Transformación a cláusulas

x Conjuntos de cláusulas

u El valor de un conjunto de cláusulas S en una interpretación I es

I(S) =

{
1, si para toda C ∈ S, I(C) = 1

0, en caso contrario.

u Una interpretación I es modelo de un conjunto de cláusulas S si I(S) = 1.

u Un conjunto de cláusulas es inconsistente si no tiene modelos.

u Una fórmula F y un conjunto de cláusulas S son equivalentes si I(F ) = I(S) para

cualquier interpretación I.

u Si S = {C1, C2, . . . , Cn}, entonces S es equivalente a C1 ∧ C2 ∧ . . . ∧ Cn.

Transformación a cláusulas

x Transformación de fórmulas en FNC a conjunto de cláusulas

u El siguiente procedimiento transforma fórmulas en forma normal conjuntiva en

conjuntos cláusulas equivalentes. Cláusulas–FNC(F ) es

· Cláusulas–FNC(F1) ∪ Cláusulas–FNC(F2), si F = F1 ∧ F2

· {Cláusula(F )}, en caso contrario

u clausulas FNC(+F,?S) se verifica si S es un conjunto de cláusulas equivalente a la

fórmula en forma normal conjuntiva F. Por ejemplo,

?- clausulas_FNC(p & (-q v r), S). => S = [[p], [r, -q]]
?- clausulas_FNC((-p v q) & (-p v -r), S). => S = [[q, -p], [-p, -r]]

u Def. de clausulas FNC:

clausulas_FNC(A1 & A2, S) :- !,
clausulas_FNC(A1, S1),
clausulas_FNC(A2, S2),
union(S1, S2, S).

clausulas_FNC(A, [S]) :-
clausula(A, S).

Transformación a cláusulas

x Transformación de fórmulas a conjunto de cláusulas

u clausulas(+F,?S) se verifica si S es un conjunto de cláusulas equivalente a la fórmula

F.

u Ejemplos:

?- clausulas(p & (q => r),S).
S = [[p], [r, -q]]

u Def. de clausulas:

clausulas(F,S) :-
fnc(F,F1),
clausulas_FNC(F1,S).

Transformación a cláusulas

x Transformación de conjuntos de fórmulas a conjunto de cláusulas

u El conjunto de fórmulas S1 y el conjunto de cláusulas S2 son equivalentes si

I(S1) = I(S2) para cualquier interpretación I.

u clausulas conjunto(+CF,-CC) se verifica si CC es un conjunto de cláusulas equivalente

al conjunto de fórmulas CF.

u Ejemplo:

?- clausulas_conjunto([p => q, q => r],CC).
CC = [[q, -p], [r, -q]]

u Def. de clausulas conjunto:

clausulas_conjunto([],[]).
clausulas_conjunto([F|CF],CC) :-

clausulas(F,CC1),
clausulas_conjunto(CF,CC2),
union(CC1,CC2,CC).

x Chang, C.–L. y Lee, R.C.–T. Symbolic Logic and Mechanical Theorem
Proving (Academic Press, 1973)

x Fitting, M. First-Order Logic and Automated Theorem Proving (2nd
ed.) (Springer, 1995)

